

x86 32-bit nb302 single loop original code

.nb302_single_loop:

```
mov  edx, [esp + nb302_innerjjnr]    ;# pointer to jjnr[k]
mov  eax, [edx]
add  dword ptr [esp + nb302_innerjjnr], 4

mov  esi, [ebp + nb302_pos]
lea  eax, [eax + eax*2]

;# fetch j coordinates
xorps xmm3, xmm3
xorps xmm4, xmm4
xorps xmm5, xmm5

movss xmm3, [esi + eax*4]    ;# jxO - - -
movss xmm4, [esi + eax*4 + 4]    ;# jyO - - -
movss xmm5, [esi + eax*4 + 8]    ;# jzO - - -

movlps xmm6, [esi + eax*4 + 12]    ;# xmm6 = jxH1 jyH1 - - -
movss  xmm7, [esi + eax*4 + 20]    ;# xmm7 = jzH1 - - -
movhps xmm6, [esi + eax*4 + 24]    ;# xmm6 = jxH1 jyH1 jxH2 jyH2
movss  xmm2, [esi + eax*4 + 32]    ;# xmm2 = jzH2 - - -

;# have all coords, time for some shuffling.

shufps xmm6, xmm6, 216 ;# constant 11011000    ;# xmm6 = jxH1 jxH2 jyH1 jyH2
unpcklps xmm7, xmm2    ;# xmm7 = jzH1 jzH2 - - -
movaps  xmm0, [esp + nb302_ixO]
movaps  xmm1, [esp + nb302_jyO]
movaps  xmm2, [esp + nb302_jzO]
movlhps xmm3, xmm6    ;# xmm3 = jxO 0 jxH1 jxH2
shufps  xmm4, xmm6, 228 ;# constant 11100100    ;# xmm4 = jyO 0 jyH1 jyH2
shufps  xmm5, xmm7, 68  ;# constant 01000100    ;# xmm5 = jzO 0 jzH1 jzH2

;# store all j coordinates in jO
movaps [esp + nb302_jxO], xmm3
movaps [esp + nb302_jyO], xmm4
movaps [esp + nb302_jzO], xmm5
subps  xmm0, xmm3
subps  xmm1, xmm4
subps  xmm2, xmm5
movaps [esp + nb302_dxOO], xmm0
movaps [esp + nb302_dyOO], xmm1
movaps [esp + nb302_dzOO], xmm2
mulps  xmm0, xmm0
mulps  xmm1, xmm1
mulps  xmm2, xmm2
addps  xmm0, xmm1
addps  xmm0, xmm2    ;# have rsq in xmm0
```

```

;# do invsqrt
rsqrtps xmm1, xmm0
movaps xmm2, xmm1
mulps xmm1, xmm1
movaps xmm3, [esp + nb302_three]
mulps xmm1, xmm0
subps xmm3, xmm1
mulps xmm3, xmm2
mulps xmm3, [esp + nb302_half] ;# rinv iO - j water

movaps xmm1, xmm3
mulps xmm1, xmm0 ;# xmm1=r
movaps xmm0, xmm3 ;# xmm0=rinv
mulps xmm1, [esp + nb302_tsc]

movhlps xmm2, xmm1
cvttps2pi mm6, xmm1
cvttps2pi mm7, xmm2 ;# mm6/mm7 contain lu indices
cvtpi2ps xmm3, mm6
cvtpi2ps xmm2, mm7
movlhps xmm3, xmm2
subps xmm1, xmm3 ;# xmm1=eps
movaps xmm2, xmm1
mulps xmm2, xmm2 ;# xmm2=eps2
pslld mm6, 2
pslld mm7, 2

movd ebx, mm6
movd ecx, mm7
psrlq mm7, 32
movd edx, mm7 ;# table indices in ebx,ecx,edx

mov esi, [ebp + nb302_VFtab]

movlps xmm5, [esi + ebx*4]
movlps xmm7, [esi + ecx*4]
movhps xmm7, [esi + edx*4] ;# got half coulomb table
movaps xmm4, xmm5
shufps xmm4, xmm7, 136 ;# constant 10001000
shufps xmm5, xmm7, 221 ;# constant 11011101

movlps xmm7, [esi + ebx*4 + 8]
movlps xmm3, [esi + ecx*4 + 8]
movhps xmm3, [esi + edx*4 + 8] ;# other half of coulomb table
movaps xmm6, xmm7
shufps xmm6, xmm3, 136 ;# constant 10001000
shufps xmm7, xmm3, 221 ;# constant 11011101
;# coulomb table ready, in xmm4-xmm7
mulps xmm6, xmm1 ;# xmm6=Geps
mulps xmm7, xmm2 ;# xmm7=Heps2
addps xmm5, xmm6

```

```
addps xmm5, xmm7 ;# xmm5=Fp
mulps xmm7, [esp + nb302_two] ;# two*Heps2
```

```
xorps xmm3, xmm3
;# fetch charges to xmm3 (temporary)
movss xmm3, [esp + nb302_qqOO]
movhps xmm3, [esp + nb302_qqOH]
```

```
addps xmm7, xmm6
addps xmm7, xmm5 ;# xmm7=FF
mulps xmm5, xmm1 ;# xmm5=eps*Fp
addps xmm5, xmm4 ;# xmm5=VV
mulps xmm5, xmm3 ;# vcoul=qq*VV
mulps xmm3, xmm7 ;# fijC=FF*qq
;# at this point xmm5 contains vcoul and xmm3 fijC
```

```
addps xmm5, [esp + nb302_vctot]
movaps [esp + nb302_vctot], xmm5
xorps xmm2, xmm2
mulps xmm3, [esp + nb302_tsc]
```

```
subps xmm2, xmm3
mulps xmm0, xmm2
```

```
movaps xmm1, xmm0
movaps xmm2, xmm0
```

```
mulps xmm0, [esp + nb302_dxOO]
mulps xmm1, [esp + nb302_dyOO]
mulps xmm2, [esp + nb302_dzOO]
;# initial update for j forces
xorps xmm3, xmm3
xorps xmm4, xmm4
xorps xmm5, xmm5
subps xmm3, xmm0
subps xmm4, xmm1
subps xmm5, xmm2
movaps [esp + nb302_fjxO], xmm3
movaps [esp + nb302_fjyO], xmm4
movaps [esp + nb302_fjzO], xmm5
addps xmm0, [esp + nb302_fixO]
addps xmm1, [esp + nb302_fiyO]
addps xmm2, [esp + nb302_fizO]
movaps [esp + nb302_fixO], xmm0
movaps [esp + nb302_fiyO], xmm1
movaps [esp + nb302_fizO], xmm2
```

```
;# done with i O Now do i H1 & H2 simultaneously first get i particle coords:
movaps xmm0, [esp + nb302_ixH1]
movaps xmm1, [esp + nb302_iyH1]
```

```

movaps xmm2, [esp + nb302_izH1]
movaps xmm3, [esp + nb302_ixH2]
movaps xmm4, [esp + nb302_iyH2]
movaps xmm5, [esp + nb302_izH2]
subps  xmm0, [esp + nb302_jxO]
subps  xmm1, [esp + nb302_jyO]
subps  xmm2, [esp + nb302_jzO]
subps  xmm3, [esp + nb302_jxO]
subps  xmm4, [esp + nb302_jyO]
subps  xmm5, [esp + nb302_jzO]
movaps [esp + nb302_dxH1O], xmm0
movaps [esp + nb302_dyH1O], xmm1
movaps [esp + nb302_dzH1O], xmm2
movaps [esp + nb302_dxH2O], xmm3
movaps [esp + nb302_dyH2O], xmm4
movaps [esp + nb302_dzH2O], xmm5
mulps xmm0, xmm0
mulps xmm1, xmm1
mulps xmm2, xmm2
mulps xmm3, xmm3
mulps xmm4, xmm4
mulps xmm5, xmm5
addps xmm0, xmm1
addps xmm4, xmm3
addps xmm0, xmm2    ;# have rsqH1 in xmm0
addps xmm4, xmm5    ;# have rsqH2 in xmm4

;# start with H1, save H2 data
movaps [esp + nb302_rsqH2O], xmm4

;# do invsqrt
rsqrtps xmm1, xmm0
rsqrtps xmm5, xmm4
movaps  xmm2, xmm1
movaps  xmm6, xmm5
mulps  xmm1, xmm1
mulps  xmm5, xmm5
movaps  xmm3, [esp + nb302_three]
movaps  xmm7, xmm3
mulps  xmm1, xmm0
mulps  xmm5, xmm4
subps  xmm3, xmm1
subps  xmm7, xmm5
mulps  xmm3, xmm2
mulps  xmm7, xmm6
mulps  xmm3, [esp + nb302_half] ;# rinv H1 - j water
mulps  xmm7, [esp + nb302_half] ;# rinv H2 - j water

;# start with H1, save H2 data
movaps [esp + nb302_rinvH2O], xmm7

```

```

movaps xmm1, xmm3
mulps  xmm1, xmm0  ;# xmm1=r
movaps xmm0, xmm3  ;# xmm0=rinv
mulps  xmm1, [esp + nb302_tsc]

movhlps xmm2, xmm1
cvttps2pi mm6, xmm1
cvttps2pi mm7, xmm2  ;# mm6/mm7 contain lu indices
cvtpi2ps xmm3, mm6
cvtpi2ps xmm2, mm7
movlhps xmm3, xmm2
subps  xmm1, xmm3  ;# xmm1=eps
movaps xmm2, xmm1
mulps  xmm2, xmm2  ;# xmm2=eps2
pslld  mm6, 2
pslld  mm7, 2

movd ebx, mm6
movd ecx, mm7
psrlq mm7, 32
movd edx, mm7  ;# table indices in ebx,ecx,edx

movlps xmm5, [esi + ebx*4]
movlps xmm7, [esi + ecx*4]
movhps xmm7, [esi + edx*4] ;# got half coulomb table
movaps xmm4, xmm5
shufps xmm4, xmm7, 136 ;# constant 10001000
shufps xmm5, xmm7, 221 ;# constant 11011101

movlps xmm7, [esi + ebx*4 + 8]
movlps xmm3, [esi + ecx*4 + 8]
movhps xmm3, [esi + edx*4 + 8] ;# other half of coulomb table
movaps xmm6, xmm7
shufps xmm6, xmm3, 136 ;# constant 10001000
shufps xmm7, xmm3, 221 ;# constant 11011101
;# coulomb table ready, in xmm4-xmm7
mulps  xmm6, xmm1  ;# xmm6=Geps
mulps  xmm7, xmm2  ;# xmm7=Heps2
addps  xmm5, xmm6
addps  xmm5, xmm7  ;# xmm5=Fp
mulps  xmm7, [esp + nb302_two] ;# two*Heps2

xorps  xmm3, xmm3
;# fetch charges to xmm3 (temporary)
movss  xmm3, [esp + nb302_qqOH]
movhps xmm3, [esp + nb302_qqHH]

addps  xmm7, xmm6
addps  xmm7, xmm5 ;# xmm7=FF
mulps  xmm5, xmm1 ;# xmm5=eps*Fp
addps  xmm5, xmm4 ;# xmm5=VV

```

```

mulps xmm5, xmm3 ;# vcoul=qq*VV
mulps xmm3, xmm7 ;# fijC=FF*qq
;# at this point xmm5 contains vcoul and xmm3 fijC
addps xmm5, [esp + nb302_vctot]
movaps [esp + nb302_vctot], xmm5

xorps xmm1, xmm1

mulps xmm3, [esp + nb302_tsc]
mulps xmm3, xmm0
subps xmm1, xmm3

movaps xmm0, xmm1
movaps xmm2, xmm1
mulps xmm0, [esp + nb302_dxH1O]
mulps xmm1, [esp + nb302_dyH1O]
mulps xmm2, [esp + nb302_dzH1O]
;# update forces H1 - j water
movaps xmm3, [esp + nb302_fjxO]
movaps xmm4, [esp + nb302_fjyO]
movaps xmm5, [esp + nb302_fjzO]
subps xmm3, xmm0
subps xmm4, xmm1
subps xmm5, xmm2
movaps [esp + nb302_fjxO], xmm3
movaps [esp + nb302_fjyO], xmm4
movaps [esp + nb302_fjzO], xmm5
addps xmm0, [esp + nb302_fixH1]
addps xmm1, [esp + nb302_fiyH1]
addps xmm2, [esp + nb302_fizH1]
movaps [esp + nb302_fixH1], xmm0
movaps [esp + nb302_fiyH1], xmm1
movaps [esp + nb302_fizH1], xmm2
;# do table for H2 - j water interaction
movaps xmm0, [esp + nb302_rinvH2O]
movaps xmm1, [esp + nb302_rsqH2O]
mulps xmm1, xmm0 ;# xmm0=rinv, xmm1=r
mulps xmm1, [esp + nb302_tsc]

movhlps xmm2, xmm1
cvttps2pi mm6, xmm1
cvttps2pi mm7, xmm2 ;# mm6/mm7 contain lu indices
cvtpi2ps xmm3, mm6
cvtpi2ps xmm2, mm7
movlhps xmm3, xmm2
subps xmm1, xmm3 ;# xmm1=eps
movaps xmm2, xmm1
mulps xmm2, xmm2 ;# xmm2=eps2
pslld mm6, 2
pslld mm7, 2

```

```

movd ebx, mm6
movd ecx, mm7
psrlq mm7, 32
movd edx, mm7      ;# table indices in ebx,ecx,edx

movlps xmm5, [esi + ebx*4]
movlps xmm7, [esi + ecx*4]
movhps xmm7, [esi + edx*4] ;# got half coulomb table
movaps xmm4, xmm5
shufps xmm4, xmm7, 136 ;# constant 10001000
shufps xmm5, xmm7, 221 ;# constant 11011101

movlps xmm7, [esi + ebx*4 + 8]
movlps xmm3, [esi + ecx*4 + 8]
movhps xmm3, [esi + edx*4 + 8] ;# other half of coulomb table
movaps xmm6, xmm7
shufps xmm6, xmm3, 136 ;# constant 10001000
shufps xmm7, xmm3, 221 ;# constant 11011101
;# coulomb table ready, in xmm4-xmm7
mulps  xmm6, xmm1      ;# xmm6=Geps
mulps  xmm7, xmm2      ;# xmm7=Heps2
addps  xmm5, xmm6
addps  xmm5, xmm7      ;# xmm5=Fp
mulps  xmm7, [esp + nb302_two] ;# two*Heps2

xorps  xmm3, xmm3
;# fetch charges to xmm3 (temporary)
movss  xmm3, [esp + nb302_qqOH]
movhps xmm3, [esp + nb302_qqHH]

addps  xmm7, xmm6
addps  xmm7, xmm5 ;# xmm7=FF
mulps  xmm5, xmm1 ;# xmm5=eps*Fp
addps  xmm5, xmm4 ;# xmm5=VV
mulps  xmm5, xmm3 ;# vcoul=qq*VV
mulps  xmm3, xmm7 ;# fijC=FF*qq
;# at this point xmm5 contains vcoul and xmm3 fijC
addps  xmm5, [esp + nb302_vctot]
movaps [esp + nb302_vctot], xmm5

xorps  xmm1, xmm1

mulps  xmm3, [esp + nb302_tsc]
mulps  xmm3, xmm0
subps  xmm1, xmm3

movaps xmm0, xmm1
movaps xmm2, xmm1

mulps  xmm0, [esp + nb302_dxH2O]
mulps  xmm1, [esp + nb302_dyH2O]

```

```

mulps  xmm2, [esp + nb302_dzH2O]
movaps  xmm3, [esp + nb302_fjxO]
movaps  xmm4, [esp + nb302_fjyO]
movaps  xmm5, [esp + nb302_fjzO]
subps  xmm3, xmm0
subps  xmm4, xmm1
subps  xmm5, xmm2
mov  esi, [ebp + nb302_faction]
movaps  [esp + nb302_fjxO], xmm3
movaps  [esp + nb302_fjyO], xmm4
movaps  [esp + nb302_fjzO], xmm5
addps  xmm0, [esp + nb302_fixH2]
addps  xmm1, [esp + nb302_fiyH2]
addps  xmm2, [esp + nb302_fizH2]
movaps  [esp + nb302_fixH2], xmm0
movaps  [esp + nb302_fiyH2], xmm1
movaps  [esp + nb302_fizH2], xmm2

```

```

;# update j water forces from local variables
movlps  xmm0, [esi + eax*4]
movlps  xmm1, [esi + eax*4 + 12]
movhps  xmm1, [esi + eax*4 + 24]
movaps  xmm3, [esp + nb302_fjxO]
movaps  xmm4, [esp + nb302_fjyO]
movaps  xmm5, [esp + nb302_fjzO]
movaps  xmm6, xmm5
movaps  xmm7, xmm5
shufps  xmm6, xmm6, 2 ;# constant 00000010
shufps  xmm7, xmm7, 3 ;# constant 00000011
addss  xmm5, [esi + eax*4 + 8]
addss  xmm6, [esi + eax*4 + 20]
addss  xmm7, [esi + eax*4 + 32]
movss  [esi + eax*4 + 8], xmm5
movss  [esi + eax*4 + 20], xmm6
movss  [esi + eax*4 + 32], xmm7
movaps  xmm5, xmm3
unpcklps  xmm3, xmm4
unpckhps  xmm5, xmm4
addps  xmm0, xmm3
addps  xmm1, xmm5
movlps  [esi + eax*4], xmm0
movlps  [esi + eax*4 + 12], xmm1
movhps  [esi + eax*4 + 24], xmm1

```

```

dec dword ptr [esp + nb302_innerk]
jz  .nb302_updateouterdata
jmp  .nb302_single_loop

```